
Mathematical Approaches
● Differential Calculus

– Best for smooth objective functions, solvable analytically.

● Steepest Descent 
– Best for smooth, well-behaved objective functions, not (practically) 

solvable analytically.

● Conjugate Gradient
– Best for smooth – but not necessarily well-behaved – objective 

functions.

● Simplex method
– Used for linear functions, linear constraints.

● …

Linear Surface

● Smoothest/simplest possible surface for finding 
solutions.

● Objective function is linear combination of  input 
variables.

● Optima found on boundary points of  feasible region.

Non-linear Continuous Surface

● Unbroken surface, but may have sharp bends.
● Objective function includes non-linear contributions 

from input variables and interactions between them.
● Optima at peaks/low points or boundaries.



Biologically Inspired Computing 5

Discontinous Surface

● Surface broken up into many separate regions (or even 
individual points).

● Objective functions are combinations of  discontinuous 
input variables.

● Subject to “combinatorial explosion” in solution 
possibilities.

● Optima could be located virtually anywhere in the 
solution space.

● It may be acceptable to find a “good enough” solution, 
instead of  the true optimum.

Biological Examples

● Stochastic hill climbing and random search
● Pheromone trails
● Colony communication and consensus 
● …
● Evolution

Solution Techniques

● Stochastic search and hill climbing
● Simulated annealing
● Environmental reinforcement
● Swarm search
● Evolutionary algorithms

Stochastic Search & Hill Climbing
● On a real world solution surface, movement that exclusively 

follows the slope, or that stays in the vicinity of  previous 
successful searches, is susceptible to settling quickly on local 
optima (solution points that are better than any others in 
the local neighborhood) that are not global optima.

● To succeed in these conditions, random movement is mixed 
with local “greedy” behavior, to search for global optima.

● If  the slope or gradient (generalized rate of  increase in 
objective function) isn't available for a given problem type, 
other local search behavior is interspersed with random 
movement.

Simulated Annealing
● Not biologically inspired, but similar in many respects to 

stochastic hill climbing.
● Solution process uses a “temperature” parameter that is 

progressively lowered (“cooled”) as the solution 
progresses.

● New solution candidate usually constructed by a random 
step from current solution.

● Temperature controls willingness to move from a current 
solution to a worse candidate: higher temperature = 
greater willingness.

(See attached butterfly-3-sa.nlogo for an example.)

Environmental Reinforcement

● Value of  decisions taken in candidate solutions are 
stored in the environment (like pheromone trails), 
with strength scaled according to the overall value 
of  the solution.

● New solution candidates weigh the apparent value 
of  immediate decision alternatives vs. the 
“remembered” value from previous solutions.

(See attached TSP Ants.nlogo for an example.)

Swarm Search

● Many searches for optimal solutions running in 
parallel, with communication between searches.

● The path followed by an individual search is updated 
from step to step, with the motion “pulled” toward 
the best solution found so far by the swarm.

● There are similar approaches that are not biologically 
inspired – e.g. emulating the movement of  charged 
particles in an electromagnetic field.

(See “Particle Swarm Optimization” model, in NetLogo Models 
Libary, for an example.)

Evolutionary Algorithms
● Candidate solutions survive and reproduce according to fitness 

(i.e. solution value): better solutions are more likely (but not 
guaranteed) to survive and reproduce.

● Genomes of  new solutions are created by inheritance from 
one or more (usually two) parent solutions; copying errors 
(mutation) may be introduced.

● In some variants, the genotype of  an individual solution may be 
an encoding of  the decision variables (genetic algorithms) or an 
actual problem-solving/goal-seeking program (genetic 
programming).

● Other variants focus on phenotype and species-level attributes, 
rather than genomic encoding.

Challenge: Optimizing Butterfly Hill Climbing
● Rewrite the model design document for the purpose of  optimization.

– Select an appropriate optimization objective (e.g. minimize time for all to reach the 
highest point, maximize number reaching highest point).

– Assume a stochastic hill climbing approach.

– What new measures (if  any) are needed?

– What new agents, behaviors, and attributes (if  any) should be included?

● How would your model design change if  you were given
– total number of  local/global maxima?

– height of  local maxima vs. global?

– degree of  “flatness” of  the terrain?

● Alternative models for optimization
– What other biological models do you think would be effective in the butterfly hill 

climbing problem?

– Would any be more effective than stochastic hill climbing?

Optimization in the Real World

– What problems do you see on a regular basis that could 
be treated as optimization problems?

– Are there any real world situations you see on a regular 
basis that you suspect are not addressed as optimization 
problems, but should be?

Resources
● Optimization materials from 2011 kickoff

– “Optimization or Computational Solution Techniques for Mathematical Programming 
Problems”, Bennett, Robey, Robey, Brown (
http://supercomputingchallenge.org/kickoff/classes/Optimization.pdf)

● Recent Supercomputing Challenge finalist projects involving optimization
– 2012

● “Optimizing Community Detection”, Team 56, La Cueva High School 
(http://supercomputingchallenge.org/archive/11-12/finalreports/56.pdf)

● “ExcellAnts”, Team 66, Los Alamos High School 
(http://supercomputingchallenge.org/archive/11-12/finalreports/66.pdf)

● “Detection of  Alzheimer's Disease Plaques”, Team 82, Manzano High School 
(http://supercomputingchallenge.org/archive/11-12/finalreports/82.pdf)

● “Ant Colony Conundrum”, Team 118, Sat. Science & Math Academy 
(http://supercomputingchallenge.org/archive/11-12/finalreports/118.pdf) 

– 2011
● “BrilliAnts”, Team 56, Los Alamos High School

(http://supercomputingchallenge.org/archive/10-11/finalreports/56.pdf)

http://supercomputingchallenge.org/kickoff/classes/Optimization.pdf
http://supercomputingchallenge.org/kickoff/classes/Optimization.pdf
http://supercomputingchallenge.org/kickoff/classes/Optimization.pdf
http://supercomputingchallenge.org/archive/11-12/finalreports/56.pdf
http://supercomputingchallenge.org/archive/11-12/finalreports/56.pdf
http://supercomputingchallenge.org/archive/11-12/finalreports/66.pdf
http://supercomputingchallenge.org/archive/11-12/finalreports/66.pdf
http://supercomputingchallenge.org/archive/11-12/finalreports/82.pdf
http://supercomputingchallenge.org/archive/11-12/finalreports/82.pdf
http://supercomputingchallenge.org/archive/11-12/finalreports/118.pdf
http://supercomputingchallenge.org/archive/11-12/finalreports/118.pdf
http://supercomputingchallenge.org/archive/10-11/finalreports/56.pdf
http://supercomputingchallenge.org/archive/10-11/finalreports/56.pdf

globals [
  highest-patches ; Patches with elevation >= 95% of maximum elevation.
  annealing-temperature ; "temperature" that decreases from 1 to 0.
]

patches-own 
[ 
  elevation
]

;; Procedure: setup
;; Type: command
;; Agent: observer
;;
;; Initialize the model by reading the elevation of each patch from 
;; ElevationData.txt, and setting color according to elevation. Find the
;; set of patches with elevation >= 95% of the maximum. Then, create 1000
;; butterflies, randomly placed. Next, set the temperature index to 1. 
;; Finally, reset the simulation clock.

to setup  
  clear-all
  file-open "ElevationData.txt"
  while [not file-at-end?]
  [
    let next-x file-read
    let next-y file-read
    let next-elevation file-read
    ask patch next-x next-y
    [
      set elevation next-elevation
    ]
  ]
  file-close
  let min-elevation min [elevation] of patches
  let max-elevation max [elevation] of patches
  ask patches 
  [
    set pcolor scale-color green elevation min-elevation max-elevation
  ]  
  set highest-patches patches with [elevation >= 0.95 * max-elevation]
  create-turtles 100
  [
    set size 2
    setxy random-xcor random-ycor
  ]
  set annealing-temperature 1
  reset-ticks
end


;; Procedure: go
;; Type: command
;; Agent: observer
;; 
;; If the simulation clock has reached max-ticks, stop. Otherwise, ask all 
;; turtles (butterflies) to move, and update the plot. Finally, update the 
;; simulation clock.

to go
  if ticks >= max-ticks 
  [
    stop
  ]
  set annealing-temperature (1 - ticks / max-ticks)
  ask turtles 
  [
    move
  ]  
  plot count turtles-on highest-patches
  tick
end


;; Procedure: move
;; Type: command
;; Agents: turtles (butterflies)
;;
;; Select a random neighboring patch. If it is the same or higher elevation, or 
;; if a U(0, 1) random variable is lessed than a "cooling factor" (which varies 
;; from 1 to 0 over time), move to that neighbor. Otherwise, do not move.

to move
  let candidate-patch one-of neighbors
  if [elevation] of candidate-patch >= elevation 
    or random-float 1 < annealing-temperature
  [
    move-to candidate-patch
  ]
end

#Butterfly Model ODD Description
This file describes the model of Pe’er et al (2005), with some variations to demonstrate the concept of simulated annealing. The description is taken from Section 3.4 of Railsback and Grimm (2012). The file uses the markup language used by NetLogo's Info tab starting with NetLogo version 5.0.

##Purpose
The model was designed to explore questions about virtual corridors. Under what conditions do the interactions of butterfly hilltopping behavior and landscape topography lead to the emergence of virtual corridors, that is, relatively narrow paths along which many butterflies move? How does variability in the butterflies’ tendency to move uphill affect the emergence of virtual corridors?

##Entities, State Variables, and Scales
The model has two kinds of entities: butterflies and square patches of land. The patches make up a square grid landscape of 150 × 150 patches, and each patch has one state variable: its elevation. Butterflies are characterized only by their location, described as the patch they are on. Therefore, butterfly locations are in discrete units, the x- and y- coordinates of the center of their patch. Patch size and the length of one time step in the simulation are not specified because the model is generic, but when real landscapes are used, a patch corresponds to 25 × 25 m<sup>2</sup>. Simulations last for 1000 time steps; the length of one time step is not specified but should be about the time it takes a butterfly to move 25–35 m (the distance from one cell to one of its neighbor cells).

##Process Overview and Scheduling
There is only one process in the model: movement of the butterflies. On each time step, each butterfly moves once. The order in which the butterflies execute this action is unimportant because there are no interactions among the butterflies.

##Design Concepts
The _basic principle_ addressed by this model is the concept of virtual corridors—pathways used by many individuals when there is nothing particularly beneficial about the habitat in them. This concept is addressed by seeing when corridors _emerge_ from two parts of the model: the adaptive movement behavior of butterflies and the landscape they move through. This _adaptive behavior_ is modeled via a simple empirical rule that reproduces the behavior observed in real butterflies: moving uphill. This behavior is based on the understanding (not included in the model) that moving uphill leads to mating, which conveys fitness (success at passing on genes, the presumed ultimate objective of organisms). Because the hilltopping behavior is assumed a priori to be the objective of the butterflies, the concepts of _Objectives_ and _Prediction_ are not explicitly considered. There is no _learning_ in the model.

_Sensing_ is important in this model: butterflies are assumed able to identify which of the surrounding patches has the highest elevation, but to use no information about elevation at further distances. (The field studies of Pe’er 2003 addressed this question of how far butterflies sense elevation differences.)

The model does not include _interaction_ among butterflies; in field studies, Pe’er (2003) found that real butterflies do interact (they sometimes stop to visit each other on the way uphill) but decided it is not important to include interaction in a model of virtual corridors.

_Stochasticity_ is used to represent two sources of variability in movement that are too complex to represent mechanistically. Real butterflies do not always move directly uphill, likely because of (1) limits in the ability of the butterflies to sense the highest area in their neighborhood, and (2) factors other than topography (e.g., flowers that need investigation along the way) that influence movement direction. This variability is represented by assuming butterflies do not move uphill every time step; sometimes they move randomly instead. Whether a butterfly moves directly uphill or randomly at any time step is modeled stochastically, using a parameter _q_ that is the probability of an individual moving directly uphillinstead of randomly.

To allow _observation_ of virtual corridors, we will define a specific “corridor width” measure that characterizes the width of a butterfly’s path from its starting patch to a hilltop.

##Initialization
The topography of the landscape (the elevation of each patch) is initialized when the model starts. Two kinds of landscapes are used in different versions of the model: (1) a simple artificial topography, and (2) the topography of a real study site, imported from a file containing elevation values for each patch. The butterflies are initialized by creating five hundred of them and setting their initial location to a single patch or small region.

##Input Data
The environment is assumed to be constant, so the model has no input data.

##Submodels
The movement submodel defines exactly how butterflies decide whether to move uphill or randomly. Since this model applies simulated annealing, butterflies don't seek the highest patch in the neighborhood; instead, they select a random destination in the neighborhood, and move there or not, based on whether it is higher in elevation &ndash; and if not, on a comparison between a U(0, 1) random variable and a "temperature" parameter. At the very start of execution, the temperature is 1.0, so moves to lower elevations are always allowed; as the simulation proceeds, the temperature drops to 0.0, and it is increasingly unlikely for downhill moves to be accepted.


## CREDITS AND REFERENCES
Pe’er, G., Saltz, D. & Frank, K. 2005. Virtual corridors for conservation management. _Conservation Biology_, 19, 1997–2003.

Pe’er, G. 2003. Spatial and behavioral determinants of butterfly movement patterns in topographically complex landscapes. Ph.D. thesis, Ben-Gurion University of the Negev.

Railsback, S. & Grimm, V. 2012. _Agent-based and individual-based modeling: A practical introduction_. Princeton University Press, Princeton, NJ.

globals [
    objective-q
    best-tour-length
    best-tour
    current-best-tour-length
    current-best-tour
    best-ant
    average-tour-length
    initial-pheromone-level
    point-color
    point-label-color
    point-size
    ant-color
    ant-size
    current-best-tour-color
    link-color
    best-tour-color
    average-tour-color
]

breed [ants ant]

breed [points point]

ants-own [
    starting-point
    visited-points
    transited-links
    tour-length
    current-point
]

links-own [
    pheromone-level
    weight
]

to setup-globals
    set point-color yellow
    set point-label-color blue
    set point-size 5
    set ant-color red
    set ant-size 8
    set best-tour-color (green + 2)
    set current-best-tour-color (green - 3)
    set link-color (gray - 4)
    set average-tour-color blue
    set initial-pheromone-level 0.01
    set-default-shape ants "bug"
    set-default-shape points "circle"
end

to startup
  clear-all
  setup-globals
  reset-ticks
end

to setup-random-points
    no-display
    startup
    create-points number-points [
        initialize-point random-xcor random-ycor
    ]
    reset
    display
end

to import-point-list
    no-display
    let source user-file
    if (not is-boolean? source) [
        file-open source
        if (not file-at-end?) [
            let point-locations file-read
            if (is-list? point-locations) [
                startup
                foreach point-locations [
                    create-points 1 [
                        initialize-point (first ?) (last ?)
                    ]
                ]
                reset
            ]
        ]
        file-close
    ]
    display
end

to import-tsp
    no-display
    let source user-file
    if (not is-boolean? source) [
        let point-locations []
        let in-data-read-mode? false
        let eof-tag "EOF"
        let node-section-tag "NODE_COORD_SECTION"
        file-open source
        startup
        while [not file-at-end?] [
            let input-line file-read-line
            ifelse (input-line = node-section-tag) [
                set in-data-read-mode? true
            ] [
                ifelse (input-line = eof-tag) [
                    set in-data-read-mode? false
                ] [
                    if (in-data-read-mode?) [
                        let input-list (read-from-string (word "[" input-line "]"))
                        if (is-number? first input-list) [
                            set input-list (butfirst input-list)
                            set point-locations (fput input-list point-locations)
                        ]
                    ]
                ]
            ]
        ]
        file-close
        if (is-list? point-locations) [
            let min-x (min map [first ?] point-locations)
            let max-x (max map [first ?] point-locations)
            let min-y (min map [last ?] point-locations)
            let max-y (max map [last ?] point-locations)
            if ((min-x < min-pxcor - 0.5) or (max-x >= max-pxcor + 0.5)
                    or (min-y < min-pycor - 0.5) or (max-y >= max-pycor + 0.5)) [
                let data-x-range (max-x - min-x)
                let data-y-range (max-y - min-y)
                show (list (list min-x min-y) (list max-x max-y))
                let scaling-factor (min list ((world-width - 1) / data-x-range) 
                    ((world-height - 1) / data-y-range))
                let x-offset ((world-width - 1 + data-x-range * scaling-factor) / 2)
                let y-offset ((world-height - 1 + data-y-range * scaling-factor) / 2)
                set point-locations (map [list ((first ? - min-x) * scaling-factor + x-offset)
                    ((last ? - min-y) * scaling-factor + y-offset)] point-locations)
            ]
            set point-locations (reverse point-locations)
            startup            
            foreach point-locations [
                create-points 1 [
                    initialize-point (first ?) (last ?)
                ]
            ]
            reset
        ]
    ]
    display
end

to export-point-list
    if ((count points) = 0) [
        user-message "There are no points to export."
        stop
    ]
    let target user-new-file
    if (not is-boolean? target) [
        if (file-exists? target) [
            file-delete target
        ]
        file-open target
        file-print "["
        foreach (sort points) [
            file-print (word "\t[" ([xcor] of ?) "\t" ([ycor] of ?) "]")
        ]
        file-print "]"
        file-close
    ]
end

to export-tsp
    if ((count points) = 0) [
        user-message "There are no points to export."
        stop
    ]
    let tsp-title (user-input "Provide a name for the TSP data set")    
    if (not is-boolean? tsp-title) [
        let target user-new-file
        if (not is-boolean? target) [
            if (file-exists? target) [
                file-delete target
            ]
            let header 
                (list
                    (word "NAME: " tsp-title)
                    "TYPE: TSP"
                    "COMMENT: Exported from TSP Ants NetLogo model"
                    (word "DIMENSION: " (count points))
                    "EDGE_WEIGHT_TYPE: EUC_2D"
                    "NODE_COORD_SECTION"
                )
            file-open target
            file-print (reduce [(word ?1 "\n" ?2)] header)
            foreach (sort points) [
                file-print (word (1 + [who] of ?) " " ([xcor] of ?) " " ([ycor] of ?))
            ]
            file-print "EOF"
            file-close
        ]
    ]
end

to setup-plot
    set-current-plot-pen "Average"
    set-plot-pen-color average-tour-color
    set-current-plot-pen "Current Best"
    set-plot-pen-color current-best-tour-color
    set-current-plot-pen "Best"
    set-plot-pen-color best-tour-color
end

to initialize-point [x y]
    set color point-color
    set size point-size
    setxy x y
    set plabel-color point-label-color
    set plabel (word (who + 1) "  ")
    create-links-with (other points) [
        hide-link
        set color link-color
    ]
end

to initialize-ant
    set hidden? (not view-ant-motion?)
    set color ant-color
    set size ant-size
end

to reset
    clear-all-plots
    ask ants [
        die
    ]    
    ask links [
        hide-link
    ]
    reset-ticks
end

to setup-for-solution
    ask links [
        hide-link
        set color link-color
        set pheromone-level initial-pheromone-level
    ]
    ask points [
        hatch-ants ants-per-point [
            initialize-ant
        ]
    ]
    set best-tour-length 1000000
    set current-best-tour-length best-tour-length
    set best-ant nobody
    set best-tour nobody
    setup-plot
end

to iterate
    if ((count points) = 0) [
        user-message "There are no points defined. Use the \"Import Points\" or \"Random Points\" button to define TSP points."
        stop
    ]
    if ((count ants) = 0) [
        user-message "There are no ants to solve the TSP problem. Use the \"Setup/Reset Ants\" button to create ants."
        stop
    ]
    let tour-lengths []
    let current-best-ant nobody
    update-link-weights
    ifelse (view-ant-motion?) [
        display
        ask-concurrent (ants with [(not use-elitist-strategy?) or (self != best-ant)]) [
            show-turtle
            tour
        ]
    ] [
        no-display
        ask (ants with [(not use-elitist-strategy?) or (self != best-ant)]) [
            hide-turtle
            tour
        ]
    ]
    set tour-lengths ([tour-length] of ants)
    set current-best-ant (min-one-of ants [tour-length])
    set current-best-tour ([transited-links] of current-best-ant)
    set current-best-tour-length ([tour-length] of current-best-ant)
    set average-tour-length ifelse-value (best-ant = nobody) [
        mean tour-lengths
    ] [
        (best-tour-length + sum tour-lengths) / (count ants)
    ]
    if (current-best-tour-length < best-tour-length) [
        set best-tour-length current-best-tour-length
        set best-tour current-best-tour
        if (use-elitist-strategy?) [
            set best-ant current-best-ant
        ]
    ]
    update-pheromone-trails
    update-link-display
    update-plot
    tick
    if (not view-ant-motion?) [
        display
    ]
end

to update-pheromone-trails
    ask links [
        evaporate-pheromone-trails
    ]
    ask ants [
        leave-pheromones
    ]
end

to update-link-display
    ask links [
        ifelse (member? self best-tour) [
            set color best-tour-color
            show-link
        ] [
            ifelse (member? self current-best-tour) [
                set color current-best-tour-color
                show-link
            ] [
                hide-link
                set color link-color
            ]
        ]
    ]
end

to update-plot
    set-current-plot-pen "Average"
    plot average-tour-length
    set-current-plot-pen "Current Best"
    plot current-best-tour-length
    set-current-plot-pen "Best"
    plot best-tour-length
end

to tour
    start-tour
    while [(empty? transited-links) or (current-point != starting-point)] [
        advance-to-next-point
    ]
    set visited-points (reverse visited-points)
    set transited-links (reverse transited-links)
end

to leave-pheromones
    let amount ((10 ^ log10-total-pheromone) / tour-length)
    let index 0
    foreach transited-links [
        ask ? [
            set pheromone-level (pheromone-level + amount)
        ]
    ]
end

to start-tour
    set tour-length 0
    set starting-point (one-of points-here)
    set current-point starting-point
    set visited-points (list current-point)
    set transited-links []
end

to advance-to-next-point
    let total-weight 0
    let weights []
    let probabilities []
    let chosen-point starting-point
    let chosen-link nobody
    let other-points ([self] of (([link-neighbors] of current-point)) 
        with [not member? self [visited-points] of myself])
    if (not empty? other-points) [
        let inverse-sample 0
        let cumulative 0
        foreach other-points [
            let test-link [link-with ?] of current-point
            set weights (fput ([weight] of test-link) weights)
        ]
        set total-weight (sum weights)
        set weights (reverse weights)
        set weights (map [? / total-weight] weights)
        set inverse-sample (random-float 1.0)
        set cumulative 0.0
        while [cumulative <= inverse-sample] [
            set cumulative (cumulative + first weights)
            set weights (butfirst weights)
            set chosen-point (first other-points)
            set other-points (butfirst other-points)
        ]
    ]
    set chosen-link ([link-with chosen-point] of current-point)
    set visited-points (fput chosen-point visited-points)
    set transited-links (fput chosen-link transited-links)
    set tour-length (tour-length + [link-length] of chosen-link)
    face chosen-point
    forward distance chosen-point
    set current-point chosen-point
end

to evaporate-pheromone-trails
    ask links [
        set pheromone-level (max (list initial-pheromone-level ((1 - (evaporation-rate / 100)) * pheromone-level)))
    ]
end

to update-link-weights 
    ask links [
        set weight ((pheromone-level ^ pheromone-weight) / (link-length ^ distance-weight))
    ]
end

## WHAT IS IT?

This is a model which demonstrates the use of an "ant-based" approach to obtain a reasonably good solution to the Traveling Salesman Problem.

## HOW IT WORKS

This section could explain what rules the agents use to create the overall behavior of the model.

## HOW TO USE IT

This section could explain how to use the model, including a description of each of the items in the interface tab.

## THINGS TO NOTICE

This section could give some ideas of things for the user to notice while running the model.

## THINGS TO TRY

This section could give some ideas of things for the user to try to do (move sliders, switches, etc.) with the model.

## EXTENDING THE MODEL

This section could give some ideas of things to add or change in the procedures tab to make the model more complicated, detailed, accurate, etc.

## NETLOGO FEATURES

This section could point out any especially interesting or unusual features of NetLogo that the model makes use of, particularly in the Procedures tab.  It might also point out places where workarounds were needed because of missing features.

## RELATED MODELS

This section could give the names of models in the NetLogo Models Library or elsewhere which are of related interest.

## CREDITS AND REFERENCES

This section could contain a reference to the model's URL on the web if it has one, as well as any other necessary credits or references.

